

# Berechnung von Konfidenzintervallen für die Zahl "Number Needed to Treat" (NNT)

Ralf Bender

AG 3 – Epidemiologie und Medizinische Statistik  
Fakultät für Gesundheitswissenschaften  
Universität Bielefeld  
Telefon: 0521 / 106-3803  
eMail: Ralf.Bender@uni-bielefeld.de

## Abstract

Die Zahl "Number Needed to Treat" (NNT) wird seit einigen Jahren vor allem in der Ergebnisdarstellung randomisierter klinischer Studien immer häufiger verwendet, um die Stärke des Effekts einer neuen Behandlung zu beschreiben. Die Zahl NNT gibt die Anzahl von Patienten an, die über einen bestimmten Zeitraum eine neue anstelle einer Standardbehandlung erhalten müssen, um ein zusätzliches (ungünstiges) Ereignis zu verhindern. Da NNT der Kehrwert der absoluten Risikoreduktion (ARR) ist, besteht der mögliche Wertebereich von NNT aus den beiden Intervallen  $[1, \infty)$  und  $(-\infty, -1]$ , wobei der Wert  $\text{NNT}=1$  den theoretisch größten Nutzen,  $\text{NNT}=-1$  den theoretisch größten Schaden und  $\pm\infty$  den Nulleffekt der Behandlung darstellt. Werte zwischen  $-1$  und  $1$  können für NNT nicht vorkommen. Dies hat Konsequenzen für die Darstellung von Konfidenzintervallen für NNT. Bei nicht signifikanten Behandlungseffekten besteht das Konfidenzintervall für NNT aus 2 Bereichen, was bei der Präsentation von Studienergebnissen berücksichtigt werden muß. Die gängige Praxis ist allerdings, daß Konfidenzintervalle für NNT nur bei signifikanten Behandlungseffekten angegeben werden, was nicht zufriedenstellend ist. Hierbei werden Konfidenzintervalle für NNT über den Kehrwert der Konfidenzgrenzen für ARR berechnet. Zur Berechnung von Konfidenzintervallen für ARR wird fast ausschließlich die asymptotische Standardmethode verwendet. Diese Methode liefert jedoch bei kleinen Stichproben und kleinen Behandlungseffekten keine glaubwürdigen Resultate. Es wird daher ein SAS/IML-Programm vorgestellt, mit dem Konfidenzintervalle für NNT mit Hilfe der Wilson Score Methode berechnet werden können. Diese Methode führt vor allem bei kleinen Stichproben und kleinen Behandlungseffekten zu einer verbesserten Berechnung von Konfidenzintervallen für NNT und ist auch anwendbar in Situationen, in denen die asymptotische Standardmethode gar kein Ergebnis liefert.

## Hintergrund

Die Zahl "Number Needed to Treat" (NNT) wird seit einigen Jahren immer häufiger verwendet, um die Ergebnisse randomisierter klinischer Studien mit binärer Zielvariable darzustellen (Cook & Sackett, 1995). Definiert als Kehrwert der absoluten Risikoreduktion (ARR) gibt NNT die geschätzte Anzahl von Patienten an, die eine neue anstelle einer Standardbehandlung erhalten müssen um ein unerwünschtes Ereignis bei einem zusätzlichen Patienten zu verhindern. Ein negativer NNT-Wert ist somit die geschätzte Zahl von Patienten, die behandelt werden müssen, um einem zusätzlichen Patienten Schaden zuzufügen. Obwohl das Maß NNT von Klinikern und Patienten besser verstanden wird als Risikoquotienten oder Risikoreduktionen, hat es ungünstige mathematische und statistische Eigenschaften. Eine ausführliche Erläuterung zu Konfidenzintervallen für NNT wurde kürzlich von Altman (1998) verfaßt. Die allgemeinen mathematischen und statistischen Eigenschaften von NNT wurden im Detail von Lesaffre & Pledger (1999) beschrieben. Der Schlüssel zum Verständnis eines Konfidenzintervalls für NNT ist, daß der prinzipielle Wertebereich von NNT durch die beiden Intervalle  $[1, \infty)$  und  $(-\infty, -1]$  gegeben ist. Werte zwischen  $-1$  und  $1$  sind unmöglich für NNT. Der Wert, der den größtmöglichen nützlichen Effekt einer Behandlung wiedergibt, ist  $\text{NNT}=1$ , was bedeutet, daß 1 Patient behandelt werden muß um ein unerwünschtes Ereignis

zu verhindern. Der Wert, der keinerlei Behandlungseffekt beschreibt ( $ARR=0$ ), ist  $NNT=\pm\infty$ , und der Wert, der den größtmöglichen schädlichen Effekt wiedergibt, ist  $NNT=-1$ . Demnach bedeutet ein Wert von  $NNT=10$  mit Konfidenzintervall von 4 bis  $-20$ , daß die beiden Regionen  $[4,\infty)$  und  $(-\infty,-20]$  das Konfidenzintervall bilden, nicht etwa die Region  $[-20,4]$ . Altman (1998) legt nahe, daß in der Ergebnisdarstellung randomisierter klinischer Studien jede  $NNT$ -Schätzung durch ein entsprechendes Konfidenzintervall ergänzt werden sollte. Die gängige Praxis ist allerdings, daß Konfidenzintervalle für  $NNT$  nur bei signifikanten Behandlungseffekten angegeben werden, was nicht zufriedenstellend ist (Altman, 1998; Bender, 2000). Des weiteren kann die gewöhnliche Methode zur Berechnung dieser Konfidenzintervalle deutlich verbessert werden. Konfidenzintervalle für  $NNT$  können durch Umkehrung und Vertauschung der Konfidenzgrenzen für  $ARR$  berechnet werden (Cook & Sackett, 1995). Die einzige in der Praxis routinemäßig angewendete Methode zur Berechnung von Konfidenzintervallen für  $ARR$  ist jedoch die gewöhnliche asymptotische Methode (Altman, 1998; Daly, 1998). In der statistischen Literatur ist bekannt, daß die asymptotische Methode in vielen Fällen zu schmale Konfidenzintervalle für  $ARR$  liefert (Beal, 1987; Wallenstein, 1997; Newcombe, 1998b). Daher schlägt Newcombe (1998b) vor, eine Methode basierend auf Wilson Score Intervallen zu verwenden. Diese Methode hat deutliche Vorteile gegenüber der gewöhnlichen asymptotischen Methode und wird in dieser Arbeit angewendet.

## Methode

Seien  $\pi_1$  und  $\pi_2$  die wahren Wahrscheinlichkeiten (Risiken) eines unerwünschten Ereignisses in der Kontrollgruppe (Gruppe 1) und in der Behandlungsgruppe (Gruppe 2). Die wahre absolute Risikoreduktion ist die Differenz der beiden Risiken  $ARR=\pi_1-\pi_2$ . Die wahre Zahl  $NNT$  ist die Umkehrung von  $ARR$ , d.h.  $NNT=1/ARR=1/(\pi_1-\pi_2)$ . Zur Schätzung dieser Maße kann ein randomisierter klinischer Versuch durchgeführt werden. Seien  $n_1$  und  $n_2$  die Anzahlen der Patienten, die in die Kontrollgruppe bzw. in die Behandlungsgruppe randomisiert werden und  $e_1$  und  $e_2$  die Anzahlen der Patienten in der Kontrollgruppe bzw. in der Behandlungsgruppe mit einem unerwünschten Ereignis. Die beiden Risiken können dann geschätzt werden durch  $p_1=e_1/n_1$  und  $p_2=e_2/n_2$ . Eine Schätzung für die absolute Risikoreduktion ist gegeben durch  $ARR=p_1-p_2$  und  $NNT$  kann geschätzt werden durch  $NNT=1/(p_1-p_2)$ . Ein approximatives Konfidenzintervall für  $NNT$  kann berechnet werden durch Umkehrung und Vertauschung der Konfidenzgrenzen für  $ARR$ . Daher konzentrieren wir uns auf die Berechnung von Konfidenzintervallen für  $ARR$ . Die Standardmethode um Konfidenzintervalle für  $ARR$  zu berechnen, basiert auf der asymptotischen Normalverteilung und der gewöhnlichen Formel für den Standardfehler der geschätzten  $ARR$  (Daly, 1998).

Während die asymptotische Methode bei großem Stichprobenumfang und hohen  $ARR$  Werten zufriedenstellend ist, liefert sie in vielen Fällen zu schmale Konfidenzintervalle (Beal, 1987; Wallenstein, 1997; Newcombe, 1998b). Dennoch werden in der Praxis – wenn überhaupt – Konfidenzintervalle für  $ARR$  und  $NNT$  mit Hilfe der asymptotischen Methode berechnet. Exakte Konfidenzintervalle für  $ARR$ , die z.B. mit Hilfe von StatXact berechnet werden können, sind konservativ, d.h. unnötig breit (Agresti & Coull, 1998). Es wurde häufig gezeigt, daß Konfidenzintervalle basierend auf Wilson Scores Überdeckungswahrscheinlichkeiten besitzen, die sehr nahe am nominalen Konfidenzniveau liegen (Agresti & Coull, 1998; Vollset, 1993; Newcombe, 1998a; Newcombe, 1998b). Außerdem sind diese viel leichter zu berechnen als exakte Konfidenzintervalle, da lediglich die Lösung einer quadratischen Gleichung notwendig ist. Nach einem Vergleich von 11 Methoden zur Berechnung von Konfidenzintervallen für  $ARR$  legt Newcombe (1998b) nahe, die Wilson Score Methode zu verwenden. Die expliziten Formeln zur Berechnung von Konfidenzintervallen für  $ARR$  und

NNT basierend auf Wilson Scores werden bei Bender (2000) angegeben. Zur Berechnung wurde ein SAS/IML-Programm (SAS, 1985) erstellt, das im Anhang aufgeführt ist.

## Vergleich der asymptotischen und der Wilson Score Methode

Die gewöhnliche asymptotische Methode zur Berechnung von Konfidenzintervallen für ARR hat schlechte Überdeckungseigenschaften und kann irreguläre Werte liefern. Speziell bei kleinen Stichproben und ARR Werten nahe an 0 oder 1 führt die asymptotische Methode zu ungläubwürdigen Resultaten. Für geschätzte ARR Werte von 0 oder 1 liefert die asymptotische Methode kein brauchbares Konfidenzintervall. Diese Nachteile haben beträchtliche Bedeutung in Studien mit niedrigem Behandlungseffekt und in Äquivalenzstudien. In diesen Situationen, in denen es besonders wichtig ist die Unsicherheit von Schätzungen zu quantifizieren, versagt die asymptotische Methode. Im folgenden werden anhand von künstlichen Beispielen die asymptotischen mit den Wilson Score Konfidenzintervallen verglichen.

**Tab. 1:** Konfidenzintervalle für NNT Werte berechnet mit der asymptotischen und der Wilson Score Methode bei künstlichen Beispielen

	Beschreibung der Beispiele	Kontrollgruppe			Behandlungsgruppe			ARR	NNT	95% KI	
		e <sub>1</sub>	n <sub>1</sub>	p <sub>1</sub>	e <sub>2</sub>	n <sub>2</sub>	p <sub>2</sub>			as.	Wilson
1	as. OG ungläubwürdig (zu niedrig)	10	200	0.050	3	200	0.015	0.035	28.6	14 bis 2411	13 bis -890
2	as. UG theoretisch unmöglich (<1)	6	7	0.857	1	7	0.143	0.714	1.4	0.9 bis 2.9	1.1 bis 5.2
3	kein brauchbares as. KI falls ARR=1	5	5	1.000	0	5	0.000	1.000	1.0	1.0 bis 1.0	1.0 bis 2.6
4	kein brauchbares as. KI falls p <sub>1</sub> =p <sub>2</sub> =0	0	100	0.000	0	100	0.000	0.000	∞	∞ bis ∞	27 bis -27
5	as. KI inadäquat zur Prüfung von Äquivalenz	1	500	0.002	2	500	0.004	-0.002	-500	209 bis -114	130 bis -79

Abkürzungen: as. = asymptotisch, ARR = absolute Risikoreduktion, NNT = Number Needed to Treat, OG = obere Grenze, UG = untere Grenze, KI = Konfidenzintervall

Die Nachteile der asymptotischen Methode werden durch die Beispiele aus Tabelle 1 herausgestellt. Für hohe NNT Werte und mittlerem Stichprobenumfang ist insbesondere die obere asymptotische Konfidenzgrenze ungläubwürdig (Bsp. 1). Bei niedrigen NNT Werten und kleiner Stichprobe kann es eine theoretisch unmögliche untere Konfidenzgrenze geben (Bsp. 2). Bei einer ARR Schätzung von genau 1 ergibt sich ein unbrauchbares Konfidenzintervall, da der Standardfehler irrtümlich 0 ist (Bsp. 3). Das gleiche gilt wenn beide geschätzten Risiken genau 0 sind (Bsp. 4).

Die ungläubwürdigen Resultate der asymptotischen Methode sind insbesondere für Äquivalenzstudien von Bedeutung. Eine gebräuchliche Methode zum Nachweis der Äquivalenz ist zu zeigen, daß das 95% Konfidenzintervall des gewählten Maßes komplett im vorher definierten Äquivalenzbereich enthalten ist. Ein möglicher Äquivalenzbereich für NNT könnte beispielsweise das Intervall von 100 bis -100 sein. Dies würde bedeuten, daß zwei Behandlungen äquivalent sind, wenn 100 oder mehr Patienten behandelt werden müssen, um 1 zusätzlichen Patienten zu nützen oder zu schaden. Wenn beide Behandlungen sehr effektiv sind um Patienten vor ungünstigen Ereignissen zu bewahren, ist die Zahl der

beobachteten Ereignisse gering. Falls die Zahl der beobachteten Ereignisse in beiden Gruppen sogar 0 ist, erhält man mit der asymptotischen Methode überhaupt kein brauchbares Konfidenzintervall, was bedeutet, daß gar keine Entscheidung möglich ist (Bsp. 4). Falls die Zahl der beobachteten Ereignisse in beiden Gruppen niedrig ist, führt die asymptotische Methode zu unglaublichen Konfidenzintervallen selbst wenn der Stichprobenumfang groß ist. In Beispiel 5 würde das asymptotische Konfidenzintervall von 209 bis  $-114$  zu der Entscheidung für Äquivalenz führen, da es komplett im Äquivalenzbereich von 100 bis  $-100$  enthalten ist. Diese Entscheidung ist jedoch fragwürdig, da das asymptotische Konfidenzintervall zu schmal ist. Dies zeigt das Wilson Score Konfidenzintervall von 130 bis  $-79$ . Das bedeutet, daß bis zu 1 von 79 Patienten geschadet werden kann anstelle von 1 von 100 Patienten. Also übersteigt die obere Konfidenzgrenze die definierte Äquivalenzgrenze von  $NNT=-100$ . Demzufolge sollte in Äquivalenzstudien, in denen NNT als Effektmaß verwendet wird, die asymptotische Methode nicht zur Berechnung von Konfidenzintervallen benutzt werden, selbst wenn der Stichprobenumfang hoch ist.

## Schlußfolgerung

In der aktuellen medizinischen Literatur ist die Berechnung und Präsentation von Konfidenzintervallen für NNT sehr unbefriedigend. Die einzige routinemäßig angewandte Methode ist die Umkehrung und Vertauschung der asymptotischen Konfidenzgrenzen für die absolute Risikoreduktion. Dieses Verfahren liefert jedoch häufig unglaubliche bzw. unbrauchbare Konfidenzintervalle für NNT, besonders in Studien mit niedrigem Stichprobenumfang, niedrigem Effekt und in Äquivalenzstudien. Die Anwendung der Wilson Score Methode führt zu Konfidenzintervallen für NNT, die bessere Überdeckungseigenschaften besitzen, frei sind von irregulären Werten und einfacher zu berechnen sind als exakte Konfidenzintervalle. Jede NNT Schätzung sollte ergänzt werden durch ein adäquates Konfidenzintervall. Es wird empfohlen die gewöhnliche asymptotische Methode zur Berechnung von Konfidenzintervallen für NNT durch die Wilson Score Methode oder eine andere Methode mit adäquaten Überdeckungseigenschaften zu ersetzen (Newcombe, 1998b). Zur Berechnung von Konfidenzintervallen für NNT basierend auf der Wilson Score Methode, kann das im Anhang angegebene SAS/IML-Programm verwendet werden.

## Literatur

- Agresti, A. & Coull, B.A. (1998): Approximate is better than "exact" for interval estimation of binomial proportions. *Am. Statistn.* 52, 119-126.
- Altman, D.G. (1998): Confidence intervals for the number needed to treat. *BMJ* 317, 1309-1312.
- Beal, S.L. (1987): Asymptotic confidence intervals for the difference between binomial parameters for the use with small samples. *Biometrics* 43, 941-950.
- Bender, R. (2000): Calculating confidence intervals for the number needed to treat. *Contr. Clin. Trials* (submitted for publication).
- Cook, R.J. & Sackett, D.L. (1995): The number needed to treat: A clinically useful measure of treatment effect. *BMJ* 310, 452-454.
- Daly, L.E. (1998): Confidence limits made easy: Interval estimation using a substitution method. *Am. J. Epidemiol.* 147, 783-790.

- Lesaffre, E. & Pledger, G. (1999): A note on the number needed to treat. *Contr. Clin. Trials* 20, 439-447.
- Newcombe, R.G. (1998a): Two-sided confidence intervals for the single proportion: Comparison of seven methods. *Statist. Med.* 17, 857-872.
- Newcombe, R.G. (1998b): Interval estimation for the difference between independent proportions: Comparison of eleven methods. *Statist. Med.* 17, 873-890.
- SAS (1985): *SAS/IML User's Guide, Version 5 Edition*. SAS Institute Inc., Cary, NC.
- Vollset, S.E. (1993): Confidence intervals for a binomial proportion. *Statist. Med.* 12, 809-824.
- Wallenstein, S. (1997): A non-iterative accurate asymptotic confidence interval for the difference between two proportions. *Statist. Med.* 16, 1329-1336.

## Anhang

Im folgenden ist der SAS Code des SAS/IML-Macros "nnt\_ci.sas" zur Berechnung von Konfidenzintervallen für NNT aufgeführt. Das Programm ist auch im Internet erhältlich unter "[http://www.uni-bielefeld.de/~rbender/SOFTWARE/nnt\\_ci.sas](http://www.uni-bielefeld.de/~rbender/SOFTWARE/nnt_ci.sas)".

```
*****
*           CALCULATION of CONFIDENCE INTERVALS FOR NNT           *
*           VIA CONFIDENCE INTERVALS FOR ARR                     *
*           METHOD BASED ON WILSON SCORE INTERVALS WITHOUT CONTINUITY CORRECTION *
*           REFERENCE: NEWCOMBE (1998), STATISTICS IN MEDICINE 17, 873-890 *
*****;

*-----*
*           QUESTIONS ?
*-----*
---> Dr. Ralf BENDER   Phone: +49 521 106-3803
      Fax:             +49 521 106-6465
      E-Mail:          Ralf.Bender@uni-bielefeld.de
*-----*

!!           NOTE:
!! You have to enter the actual results of the two groups
!! in lines 29 to 32.
!! e1 is the number of adverse events in the high risk group
!! (e.g. standard treatment)
!! n1 is the sample size of the high risk group
!! e2 is the number of adverse events in the low risk group
!! (e.g. new treatment)
!! n2 is the sample size of the low risk group
*-----*

options linesize=95;
proc iml;
  e1 = 6; /* Number of events in high-risk group A */
  n1 = 20; /* Sample size of high risk group A */
  e2 = 1; /* Number of events in low-risk group B */
  n2 = 20; /* Sample size of low risk group B */

*-----*First Checks*-----*
if e1>n1 then do;
  print "ERROR !!!" ,,,,,
  "The number of events (e1) in the first group is larger",
  "than the sample size (n1) !!",,,
  "Please re-enter the data and try again.";
  abort;
end;
if e2>n2 then do;
  print "ERROR !!!" ,,,,,
  "The number of events (e2) in the second group is larger",
  "than the sample size (n2) !!",,,
  "Please re-enter the data and try again.";
  abort;
end;
if e1<0 | e2<0 then do;
  print "ERROR !!!" ,,,,,
  "The number of events in the first (e1) or second group (e2)",
  "is negative !!",,,
  "Please re-enter the data and try again.";
  abort;
end;
if n1<=0 | n2<=0 then do;
  print "ERROR !!!" ,,,,,
  "The sample size in the first (n1) or second group (n2)",
  "is not positive !!",,,
  "Please re-enter the data and try again.";
  abort;
end;
n = n1+n2;
if n<10 then do;
  print "SORRY !!!" ,,,,,
  "The total sample size is extremely low (n<10)!!",
  "Please use another software providing exact methods.",,
  "Sorry for any inconvenience.";
  abort;
end;

print "CONFIDENCE INTERVALS FOR THE NUMBER NEEDED TO TREAT (NNT)",,
      "Based on Wilson score intervals without continuity correction",
      "Reference: NEWCOMBE (1998): Stat. Med. 17, 873-890" , , , ;

*-----*Descriptive*-----*
start Descript;
  r1 = e1/n1;
  r2 = e2/n2;
  ARR = r1-r2;
  if ARR=0 then NNT="infinity";
  else NNT=1/ARR;
  z90 = probit(1-(0.10/2));
  z95 = probit(1-(0.05/2));
  z99 = probit(1-(0.01/2));
  print "Events and sample sizes of the groups: " e1 n1 e2 n2 ,
        "Risks of the groups and ARR: " r1 r2 ARR ,
        "Number needed to treat (NNT): " NNT , , ;
finish;

```

```

*-----*Asymptotic Method*-----*
start Asymptot;
print "Standard asymptotic method (frequently inadequate)";
al90 = ARR - z90*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
au90 = ARR + z90*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
al95 = ARR - z95*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
au95 = ARR + z95*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
al99 = ARR - z99*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
au99 = ARR + z99*SQRT((e1*(n1-e1)/n1**3)+(e2*(n2-e2)/n2**3));
nl90 = 1/au90;    nu90 = 1/al90;
nl95 = 1/au95;    nu95 = 1/al95;
nl99 = 1/au99;    nu99 = 1/al99;
print "90% confidence intervals for ARR and NNT: " al90 au90 " " nl90 nu90;
print "95% confidence intervals for ARR and NNT: " al95 au95 " " nl95 nu95;
print "99% confidence intervals for ARR and NNT: " al99 au99 " " nl99 nu99 , , , ;
finish;

*-----*Wilson Score Method*-----*
start Wilson;
print "Confidence intervals based upon Wilson score intervals";
l190 = (2*e1+z90**2) / (2*(n1+z90**2)) - SQRT(((2*e1+z90**2)/(2*(n1+z90**2)))**2-
e1**2/(n1**2+n1*z90**2));
u190 = (2*e1+z90**2) / (2*(n1+z90**2)) + SQRT(((2*e1+z90**2)/(2*(n1+z90**2)))**2-
e1**2/(n1**2+n1*z90**2));
l290 = (2*e2+z90**2) / (2*(n2+z90**2)) - SQRT(((2*e2+z90**2)/(2*(n2+z90**2)))**2-
e2**2/(n2**2+n2*z90**2));
u290 = (2*e2+z90**2) / (2*(n2+z90**2)) + SQRT(((2*e2+z90**2)/(2*(n2+z90**2)))**2-
e2**2/(n2**2+n2*z90**2));

l195 = (2*e1+z95**2) / (2*(n1+z95**2)) - SQRT(((2*e1+z95**2)/(2*(n1+z95**2)))**2-
e1**2/(n1**2+n1*z95**2));
u195 = (2*e1+z95**2) / (2*(n1+z95**2)) + SQRT(((2*e1+z95**2)/(2*(n1+z95**2)))**2-
e1**2/(n1**2+n1*z95**2));
l295 = (2*e2+z95**2) / (2*(n2+z95**2)) - SQRT(((2*e2+z95**2)/(2*(n2+z95**2)))**2-
e2**2/(n2**2+n2*z95**2));
u295 = (2*e2+z95**2) / (2*(n2+z95**2)) + SQRT(((2*e2+z95**2)/(2*(n2+z95**2)))**2-
e2**2/(n2**2+n2*z95**2));

l199 = (2*e1+z99**2) / (2*(n1+z99**2)) - SQRT(((2*e1+z99**2)/(2*(n1+z99**2)))**2-
e1**2/(n1**2+n1*z99**2));
u199 = (2*e1+z99**2) / (2*(n1+z99**2)) + SQRT(((2*e1+z99**2)/(2*(n1+z99**2)))**2-
e1**2/(n1**2+n1*z99**2));
l299 = (2*e2+z99**2) / (2*(n2+z99**2)) - SQRT(((2*e2+z99**2)/(2*(n2+z99**2)))**2-
e2**2/(n2**2+n2*z99**2));
u299 = (2*e2+z99**2) / (2*(n2+z99**2)) + SQRT(((2*e2+z99**2)/(2*(n2+z99**2)))**2-
e2**2/(n2**2+n2*z99**2));

delta90=SQRT((e1/n1-l190)**2+(u290-e2/n2)**2);  epsil90=SQRT((u190-e1/n1)**2+(e2/n2-
l290)**2);
delta95=SQRT((e1/n1-l195)**2+(u295-e2/n2)**2);  epsil95=SQRT((u195-e1/n1)**2+(e2/n2-
l295)**2);
delta99=SQRT((e1/n1-l199)**2+(u299-e2/n2)**2);  epsil99=SQRT((u199-e1/n1)**2+(e2/n2-
l299)**2);

al90 = ARR - delta90;  au90 = ARR + epsil90;
al95 = ARR - delta95;  au95 = ARR + epsil95;
al99 = ARR - delta99;  au99 = ARR + epsil99;

nl90 = 1/au90;    nu90 = 1/al90;
nl95 = 1/au95;    nu95 = 1/al95;
nl99 = 1/au99;    nu99 = 1/al99;

print "90% confidence intervals for ARR and NNT: " al90 au90 " " nl90 nu90;
print "95% confidence intervals for ARR and NNT: " al95 au95 " " nl95 nu95;
print "99% confidence intervals for ARR and NNT: " al99 au99 " " nl99 nu99 , , , ;

finish;

run Descript;
run Asymptot;
run Wilson;
quit;

```